
Application Note #2426 
 

Generating a Helical Motion Profile 
 

Background: 

It is possible to create a helical motion 
profile using any 3-axis (or more) 
controller. To accomplish this, 
command a coordinated circular move 
between 2 axes and gearing a third 
axis to the vector motion of the 
coordinated axes. A picture of the 
helical motion path is shown on the 
right. 

 
 

Program Example: 

This example shows how to implement helical motion: 
 

#HELIX  
REM VARIABLES USED:  
REM pitch encoder cts traversed/circle 
REM speed vector speed of circle 
REM ratio gear ratio 
REM angle swept angle of circle 
REM loopcnt used if angle>32000 for large traverses 
REM lastloop angle for last loop 
#SETUP 
pitch= 360 
radius= 5000 
speed= 50000 

VS speed  
ratio= pitch/(2*3.14159*radius)  
GA,,s specify vector speed as master for Z axis 
GR,,ratio  
IN "Enter the traverse distance (encoder cts):",dist 
angle=360 * dist/pitch 
loopcnt=@INT[angle/32000] 
lastloop= 0 
 
JP#MOVE,angle<32000 
lastloop=angle - (loopcnt*32000) 
angle=32000 
 
#MOVE 
cnt=0 
#LOOP 
VMXY 
CR radius,0,angle 
VE 
BGS 
AMS 
cnt=cnt+1 
JP#LOOP,cnt<loopcnt 
angle=lastloop 
JP#LOOP,(cnt < (loopcnt + 1)) & (lastloop > 0) 
EN 

In the above example, a circle is created using the X and Y axes of the controller. The 
Z axis is geared to the coordinated motion of the X and Y axes to create the helix. 
The gear ratio is determined by the pitch of the helix and the circumference of the 
circle.  
One unavoidable limitation of the circle (CR) command is that the maximum swept 
angle of a circle must be less than 32,000 degrees. As a result, the program executes 
multiple passes of #LOOP for any calculated angle greater than 32,000 degrees.  

Method to Increase Accuracy of Helical Motion 

For the highest possible accuracy, it is recommended that the gear ratio be calculated 
using the formula ratio=pitch/(2*pi*radius) on a standard calculator. The resolution 
of the controller is limited to approximately four decimal places so the ratio entered 
should be the gear ratio you calculate truncated to four decimal places. It is then 
necessary to calculate the amount of error you accumulate during the helical motion 
due to rounding so that the #CORRECT routine can compensate for it.  For more 
information about Gearing Resolution, see Application Note 3410. 
First, determine the value of the gear ratio as internally represented by the controller. 
The controller stores decimal values as fractions with a resolution of 1/65,536. To 
determine the internal value of the gear ratio, multiply the entered gear ratio by 
65,536, truncate any fractional part of the result, and divide by 65,536. This can also 
be determined directly by typing in a gear ratio (ex. GR ,,.0114) and then multiplying 
the _GRn operand by 10,000 and displaying the result (ex. MG _GRZ*10000). This 
result divided by 10,000 should give the same result as the calculation method. 
Next, determine the amount of error that results from the rounding of the gear ratio. 
To do this, calculate the distance that the geared axis will travel and compare it to the 
distance specified using dist. The actual distance traveled is equal to the internal ratio 
divided by the exact value of the ratio you calculated times dist.  Round this number 
to the nearest integer. Subtract this value from dist and enter as delta.  
There is one more parameter to calculate: interval, the number of counts it takes for 
the geared axis to accumulate one count of error. It is equal to the actual distance 
calculated above divided by delta. This value should be rounded down to the nearest 
integer.  

Example: 
The following example shows the calculations performed for the sample program 
above: 
pitch = 360 

radius = 5000 

dist = 3999 

angle = 360*dist/pitch = 3999 

EXACT RATIO = pitch/(2*PI*radius) = 0.011459155  \ Calculated value 

ratio = 0.0114  \Value entered in program 

INTERNAL RATIO:   

ratio*65536 = 747.1104 

      \ Truncate decimal part and divide by 65536 

       747/65536 = 0.011398315 

ACTUAL DISTANCE = INTERNAL RATIO/EXACT RATIO * dist 

              = 0.011398315/0.011459155 * 3999 = 3977.768 

\round to nearest integer 

              = 3978 

delta = dist-ACTUAL DISTANCE = 3999-3978 = 21 

interval = ACTUAL DISTANCE/delta = 3978/21 = 189.4286 \round down 

                     =189 


